
#

pkgname=xfce4-settings
pkgver=4.8.2
pkgrel=1
pkgdesc="Settings manager for xfce"
arch=('ppc')
license=('GPL2')
url="http://www.xfce.org/"
groups=('xfce4')
depends=('glib2' 'gtk2' 'exo' 'libxfce4ui' 'libxfce4util' 'xfconf'\
				'dbus-glib' 'libx11' 'libxrandr' 'libxcursor' 'libnotify' \
				'libxklavier' 'tango-icon-theme' 'fontconfig' 'libxi')
makedepends=('intltool' 'pkg-config' 'gettext' 'desktop-file-utils')
optdepends=('libcanberra: for sound control')
source=(http://archive.xfce.org/src/xfce/${pkgname}/4.8/${pkgname}-${pkgver}.tar.bz2)
md5sums=('a45420fea20f29265ab6785cb1d07377')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}
  # enable Tango icon theme
	sed -i 's|Rodent|Tango|g' \
			dialogs/appearance-settings/main.c \
			xfsettingsd/xsettings.xml
    
  ./configure --prefix=/usr \
	--sysconfdir=/etc \
	--localstatedir=/var \
	--libexecdir=/usr/lib/xfce4 \
	--disable-static \
	--enable-xrandr \
	--enable-xcursor \
	--enable-libnotify \
	--enable-libxklavier \
	--enable-pluggable-dialogs \
	--enable-nls
  make
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=${pkgdir} install
}
